package getl.jdbc

import com.fasterxml.jackson.annotation.JsonIgnore
import getl.exception.IOFilesError
import getl.exception.RequiredParameterError
import getl.utils.FileUtils
import groovy.transform.InheritConstructors

/**
 * Query dataset class
 * @author Alexsey Konstantinov
 *
 */
@InheritConstructors
class QueryDataset extends JDBCDataset {
	@Override
	protected void initParams() {
		super.initParams()

		sysParams.isQuery = true
	}

	/** Use specified connection */
	JDBCConnection useConnection(JDBCConnection value) {
		setConnection(value)
		return value
	}

	@Override
	@JsonIgnore
	Type getType() { super.getType()?:queryType }

	/** SQL query text */
	String getQuery () { params.query as String }
	/** SQL query text */
	void setQuery (String value) { params.query = value }

	/** The path to script file */
	String getScriptFilePath() { params.scriptFilePath as String }
	/** The path to script file */
	void setScriptFilePath(String value) { params.scriptFilePath = value }

	/** Script file text encoding */
	String getScriptFileCodePage() { params.scriptFileCodePage as String }
	/** Script file text encoding */
	void setScriptFileCodePage(String value) { params.scriptFileCodePage = value }

	@Override
	@JsonIgnore
	String getObjectName() { (dslNameObject != null)?"Jdbc query [$dslNameObject]":'Jdbc query' }

	/**
	 * Load script from file
	 * @param filePath file name sql batch file
	 * @param codePage file use specified encoding page (default utf-8)
	 */
	void loadFile(String filePath = null, String codePage = null) {
		setQuery(readFile(filePath?:scriptFilePath, codePage?:scriptFileCodePage))
	}

	/**
	 * Read script from file
	 * @param filePath file name sql batch file
	 * @param codePage file use specified encoding page (default utf-8)
	 * @return script text
	 */
	String readFile(String filePath, String codePage = null) {
		if (filePath == null)
			throw new RequiredParameterError('filePath', 'readFile')

		def file = new File(FileUtils.TransformFilePath(filePath, dslCreator))
		if (!file.exists())
			throw new IOFilesError(this, '#io.file.not_found', [path: filePath, type: 'SQL'])
		return file.getText(codePage?:scriptFileCodePage?:'utf-8')
	}

	/**
	 * Load script from file in class path or resource directory
	 * @param filePath file name in resource catalog
	 * @param otherPath the string value or list of string values as search paths if file is not found in the resource directory
	 * @param codePage file use specified encoding page (default utf-8)
	 */
	void loadResource(String filePath, def otherPath = null, String codePage = null) {
		setQuery(readFile('resource:/' + filePath, codePage))
	}
}